The I/O relation

A worksheet motivated by a parameters estimation problem

```
> restart;
  with (DifferentialAlgebra0);
```

[*Attributes*, *BelongsTo*, *Coeffs*, *DeltaPolynomial*, *DifferentialPrem*, *DifferentialRing*, *Differentiate*, *Display*, *Equations*, *EssentialComponents*, *FactorDerivative*, *FieldElement*, *Indets*, *Inequations*, *Initial*, *IsConstant*, *IsDifferentialRing*, *IsOrthonomic*, *IsReduced*, *IsRegularDifferentialChain*, *LeadingCoefficient*, *LeadingDerivative*, *LeadingRank*, *MaxRankElement*, *MinRankElement*, *NormalForm*, *Notation*, *Parameters*, *Pardi*, *PowerSeriesSolution*, *PowerSeriesSystem*, *PreparationEquation*, *PretendRegularDifferentialChain*, *Ranking*, *RatBilge*, *ReducedForm*, *RosenfeldGroebner*, *Separant*, *SortByRank*, *Tail*] **(1)**

A compartmental model x1 is observed while x2 is not.

```
> params := [k[e], V[e], k[12], k[21]];
```

$$params := [k_e, V_e, k_{12}, k_{21}] \quad \textbf{(2)}$$

```
> R := DifferentialRing (derivations = [t],
                         blocks = [[x[1],x[2]], params],
                         parameters = params,
                         notation = diff);
```

$$R := differential\_ring \quad \textbf{(3)}$$

```
> edoA := diff(x[1](t),t) = -k[12]*x[1](t) + k[21]*x[2](t) -
                                   (V[e]*x[1](t))/(k[e]+x[1](t));
  edoB := diff(x[2](t),t) = k[12]*x[1](t) - k[21]*x[2](t);
```

$$edoA := \frac{d}{dt} x_1(t) = -k_{12}\, x_1(t) + k_{21}\, x_2(t) - \frac{V_e x_1(t)}{k_e + x_1(t)}$$

$$edoB := \frac{d}{dt} x_2(t) = k_{12}\, x_1(t) - k_{21}\, x_2(t) \quad \textbf{(4)}$$

The perfect differential ideal generated by the equations, saturated by the multiplicative family generated by the denominators.
The base field is the field of the rational fractions generated by the parameters over Q.
A sophisticated way to express the fact that we do not want to discuss the possible vanishing of parameters.

```
> ideal := RosenfeldGroebner ([edoA,edoB], basefield=field
  (generators=params), R);
```

$$ideal := [regular\_differential\_chain] \quad \textbf{(5)}$$

```
> Equations (ideal [1], solved);
```

$$\left[\frac{d}{dt} x_2(t) = k_{12}\, x_1(t) - k_{21}\, x_2(t), \frac{d}{dt} x_1(t) = -\frac{k_{12}\, x_1(t)^2 - k_{21}\, x_2(t)\, x_1(t) + k_{12}\, x_1(t)\, k_e + V_e x_1(t) - k_{21}\, x_2(t)\, k_e}{k_e + x_1(t)}\right] \quad \textbf{(6)}$$

Change of ranking: eliminate the non-observed variable x2, in order to get an ODE constraining x1, its derivates and the parameters

```
> IOideal := Pardi (ideal [1],
                    ranking (derivations = [t],
                             blocks = [x[2], x[1], params]));
```

$$IOideal := regular\_differential\_chain \tag{7}$$

The sought equation

```
> IOrel := Equations (IOideal, leader=derivative(x[1](t)))[1];
```

$$\begin{aligned} IOrel := {} & \left(\frac{d^2}{dt^2} x_1(t)\right) x_1(t)^2 + 2 \left(\frac{d^2}{dt^2} x_1(t)\right) x_1(t)\, k_e + \left(\frac{d^2}{dt^2} x_1(t)\right) k_e^2 \\ & + \left(\frac{d}{dt} x_1(t)\right) x_1(t)^2 k_{12} + \left(\frac{d}{dt} x_1(t)\right) x_1(t)^2 k_{21} + 2 \left(\frac{d}{dt} x_1(t)\right) x_1(t)\, k_e k_{12} \\ & + 2 \left(\frac{d}{dt} x_1(t)\right) x_1(t)\, k_e k_{21} + \left(\frac{d}{dt} x_1(t)\right) k_e^2 k_{12} + \left(\frac{d}{dt} x_1(t)\right) k_e^2 k_{21} \\ & + \left(\frac{d}{dt} x_1(t)\right) k_e V_e + x_1(t)^2 V_e k_{21} + x_1(t)\, k_e V_e k_{21} \end{aligned} \tag{8}$$

A work in progress with F. Lemaire, M. Rosenkranz and Georg Regensburger ...
The RatBilge algorithm should help transforming differential equations into integral equations.

```
> normalized_IOrel := IOrel / Initial (IOrel, R):
> L := RatBilge (normalized_IOrel, t, R);
```

$$L := \left[\frac{x_1(t)\, V_e k_{21}}{k_e + x_1(t)}, \frac{k_{12}\, x_1(t)^2 + x_1(t)^2 k_{21} - k_e^2 k_{12} - k_e^2 k_{21} - k_e V_e}{k_e + x_1(t)}, x_1(t)\right] \tag{9}$$

The meaning of the above output

```
> L[1] + add (Diff (L[i], t$(i-1)), i = 2 .. nops (L));
```

$$\frac{x_1(t)\, V_e k_{21}}{k_e + x_1(t)} + \frac{\partial}{\partial t}\left(\frac{k_{12}\, x_1(t)^2 + x_1(t)^2 k_{21} - k_e^2 k_{12} - k_e^2 k_{21} - k_e V_e}{k_e + x_1(t)}\right) + \frac{d^2}{dt^2} x_1(t) \tag{10}$$

Just check

```
> normal (normalized_IOrel - add (Differentiate (L[i], t$(i-1), R), i = 1 .. nops (L)));
```

$$0 \tag{11}$$